
Communicating with a BASIC Stamp

The following code and schematic demonstrate how to interface the Parallax BASIC Stamp to the SpeakJet Demo Board’s DB-9 for serially-controlled speech and sound effects. The example works with any of the 24-pin Parallax BASIC Stamp modules (BS2, BS2e, BS2p and BS2SX) and automatically corrects for their different baud rates. The circuit requires connecting two BASIC Stamp I/O pins and ground to the SpeakJet’s DB-9. This code is available for download on the Parallax web site www.parallax.com where the Tigerbotics SpeakJet products are featured.

' =========================================================================

'

'   File...... SpeakJet_Demo.BS2

'   Purpose... Serial Control of TIGERBOTICS SpeakJet Board

'   Author.... Parallax, Inc.

'   E-mail.... support@parallax.com

'   Started...

'   Updated... 30 JUL 2004

'

'   {$STAMP BS2}

'   {$PBASIC 2.5}

'

' =========================================================================

' -----[ Program Description ]---------------------------------------------

'

' Connections:

'

'   P15  ---->  DB9.2

'   P14  <----  DB9.8

'   Vss  -----  DB9.5

' -----[ Revision History ]------------------------------------------------

' -----[ I/O Definitions ]-------------------------------------------------

Tx              PIN     15                      ' serial out to SpeakJet

Rdy             PIN     14                      ' Ready in from SpeakJet

' -----[ Constants ]-------------------------------------------------------

#SELECT $stamp

  #CASE BS2, BS2E, BS2PE

    T2400       CON     396

    T4800       CON     188

    T9600       CON     84

    T19K2       CON     32

  #CASE BS2SX, BS2P

    T2400       CON     1021

    T4800       CON     500

    T9600       CON     240

    T19K2       CON     110

#ENDSELECT

Inverted        CON     $4000

Baud            CON     T9600 + Inverted

Sync            CON     $55

' -----[ SpeakJet Control Codes ]------------------------------------------

PA0             CON     0                       ' pauses

PA1             CON     1

PA2             CON     2

PA3             CON     3

PA4             CON     4

PA5             CON     5

PA6             CON     6

Fast            CON     7

Slow            CON     8

Stress          CON     14

Relax           CON     15

_Wait           CON     16                      ' note underscore

Soft            CON     18

Volume          CON     20

Speed           CON     21

Pitch           CON     22

Bend            CON     23

PortCtr         CON     24

Port            CON     25

Repeat          CON     26

CallPhr         CON     28

GotoPhr         CON     29

Delay           CON     30

Reset           CON     31

IY              CON     128

IH              CON     129

EY              CON     130

EH              CON     131

AY              CON     132

AX              CON     133

UX              CON     134

OH              CON     135

AW              CON     136

OW              CON     137

UH              CON     138

UW              CON     139

MM              CON     140

NE              CON     141

NO              CON     142

NGE             CON     143

NGO             CON     144

LE              CON     145

LO              CON     146

WW              CON     147

RR              CON     148

IYRR            CON     149

EYRR            CON     150

AXRR            CON     151

AWRR            CON     152

OWRR            CON     153

EYIY            CON     154

OHIY            CON     155

OWIY            CON     156

OHIH            CON     157

IYEH            CON     158

EHLL            CON     159

IYUW            CON     160

AXUW            CON     161

IHWW            CON     162

AYWW            CON     163

OWWW            CON     164

JH              CON     165

VV              CON     166

ZZ              CON     167

ZH              CON     168

DH              CON     169

BE              CON     170

BO              CON     171

EB              CON     172

OB              CON     173

DE              CON     174

_DO             CON     175                     ' note underscore

ED              CON     176

OD              CON     177

GE              CON     178

GO              CON     179

EG              CON     180

OG              CON     181

CH              CON     182

HE              CON     183

HO              CON     184

WH              CON     185

FF              CON     186

SE              CON     187

SO              CON     188

SH              CON     189

TH              CON     190

TT              CON     191

TU              CON     192

TS              CON     193

KE              CON     194

KO              CON     195

EK              CON     196

OK              CON     197

PE              CON     198

PO              CON     199

RO              CON     200

R1              CON     201

R2              CON     202

R3              CON     203

R4              CON     204

R5              CON     205

R6              CON     206

R7              CON     207

R8              CON     208

R9              CON     209

A0              CON     210

A1              CON     211

A2              CON     212

A3              CON     213

A4              CON     214

A5              CON     215

A6              CON     216

A7              CON     217

A8              CON     218

A9              CON     219

_B0             CON     220                     ' note underscore

_B1             CON     221

_B2             CON     222

_B3             CON     223

_B4             CON     224

_B5             CON     225

_B6             CON     226

_B7             CON     227

_B8             CON     228

_B9             CON     229

C0              CON     230

C1              CON     231

C2              CON     232

C3              CON     233

C4              CON     234

C5              CON     235

C6              CON     236

C7              CON     237

C8              CON     238

C9              CON     239

D0              CON     240                     ' DTMF tones

D1              CON     241

D2              CON     242

D3              CON     243

D4              CON     244

D5              CON     245

D6              CON     246

D7              CON     247

D8              CON     248

D9              CON     249

D10             CON     250

D11             CON     251

M0              CON     252                     ' sonar pin

M1              CON     253                     ' pistol shot

M2              CON     254                     ' WOW

EOS             CON     255                     ' end of string

' -----[ Variables ]-------------------------------------------------------

eeAddr          VAR     Word                    ' address of speech element

element         VAR     Byte                    ' speech element / code

idx             VAR     Word

' -----[ EEPROM Data ]-----------------------------------------------------

' SCP strings

NoPuMsg         DATA    "\0237J32H80NX", EOS

ClearBuf        DATA    "\0RX", EOS

ClearAll        DATA    "\08J0N1J0N11J0N2J0N12J0N3J0N13J0NX", EOS

AllOff          DATA    "\08J0N11J0N12J0N13J0NX", EOS

DialTone        DATA    "\08J0N1J350N2J440N11J18N12J18NX", EOS

' allophone strings

Defaults        DATA    Reset, Volume, 127, EOS

Welcome         DATA    WW, EH, LE, PA4, KO, AW, MM, PA5, EOS

SpeakJet        DATA    SE, PE, IY, PA4, KE, PA5, JH, EY, TT, PA5, EOS

Demo            DATA    DE, EH, MM, OWWW, PA5, EOS

Parallax        DATA    PE, Fast, EYRR, UX, LE, AY, AY, KE, SE, PA5, EOS

Basic           DATA    BE, EYIY, SE, IH, Fast, PA4, OK, PA5, EOS

Stamp           DATA    SE, TU, AY, EH, MM, PE, PA5, EOS

RobotSad        DATA    Reset, Bend, 0, Speed, 61, _B8, PA1, C9, PA1,

                        Reset, EOS

ParaPhone       DATA    Slow, D9, PA5, Slow, D1, PA5, Slow, D6, PA5,

                        Slow, D6, PA5, Slow, D2, PA5, Slow, D4, PA5,

                        Slow, D8, PA5, Slow, D3, PA5, Slow, D3, PA5,

                        Slow, D3, EOS

' -----[ Initialization ]--------------------------------------------------

Setup:

  GOSUB No_PwrUp_Msg                            ' no "Ready" prompt

' -----[ Program Code ]----------------------------------------------------

Main:

  FOR idx = 0 TO 1

    LOOKUP idx, [Defaults, Welcome], eeAddr

    GOSUB Play_String

  NEXT

  PAUSE 1500

  FOR idx = 0 TO 1

    LOOKUP idx, [SpeakJet, Demo], eeAddr

    GOSUB Play_String

  NEXT

  PAUSE 750

  eeAddr = RobotSad

  GOSUB Play_String

  PAUSE 3000

  FOR idx = 0 TO 2

    LOOKUP idx, [Parallax, Basic, Stamp], eeAddr

    GOSUB Play_String

  NEXT

  PAUSE 3000

  FOR idx = 15 TO 127 STEP 16

    SEROUT Tx\Rdy, Baud, [Volume, idx]          ' set volume level

    eeAddr = Parallax                           ' say "Parallax"

    GOSUB Play_String

    PAUSE 500

  NEXT

  PAUSE 3000

  eeAddr = DialTone                             ' simulate dial tone

  GOSUB Play_String

  PAUSE 4000

  eeAddr = ParaPhone                            ' dial Parallax number

  GOSUB Play_String

  PAUSE 4000

  GOTO Main

  END

' -----[ Subroutines ]-----------------------------------------------------

' Play a string of SpeakJet sound elements from EE.  This subroutine

' uses flow-control to prevent sending elements when the SpeakJet

' buffer is full.

'

' -- put start of string in 'eeAddr'

Play_String:

  DO

    READ eeAddr, element                        ' get element from EE

    eeAddr = eeAddr + 1                         ' point to next

    IF (element = EOS) THEN EXIT                ' end? if yes we're done

    SEROUT Tx\Rdy, Baud, [element]              ' send element to SpeakJet

  LOOP

  RETURN

No_PwrUp_Msg:                                   ' clears PU Msg assignment

  eeAddr = NoPuMsg

  GOSUB Play_String

  RETURN 

Wiring Schematic:

